```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider the model of a string with an interior mass studied in H. Mounier, J. Rudolph, M. Fliess, P. Rouchon, "Tracking control of a vibrating string with an interior mass viewed as delay system", *ESAIM COCV*, 3 (1998), 315-321,

$$\begin{cases} \phi_1(t) + \psi_1(t) - \phi_2(t) - \psi_2(t) = 0, \\ \dot{\phi}_1(t) + \dot{\psi}_1(t) + \eta_1\,\phi_1(t) - \eta_1\,\psi_1(t) - \eta_2\,\phi_2(t) + \eta_2\,\psi_2(t) = 0, \\ \phi_1(t - 2\,h_1) + \psi_1(t) - u(t - h_1) = 0, \\ \phi_2(t) + \psi_2(t - 2\,h_2) - v(t - h_2) = 0, \end{cases} \tag{1}$$

where η_1, η_2 are constant parameters and $h_1, h_2 \in \mathbb{R}_+$ are such that $\mathbb{Q}\,h_1 + \mathbb{Q}\,h_2$ is a 2-dimensional $\mathbb{Q}$-vector space. Let us denote by $A = \mathbb{Q}(\eta_1, \eta_2)\,[d, \sigma_1, \sigma_2]$ the commutative polynomial algebra of differential incommensurable time-delay operators in d, σ_1 and σ_2, where:

$$d\,f(t) = \dot{f}(t), \quad \sigma_1 f(t) = f(t - h_1), \quad \sigma_2 f(t) = f(t - h_2).$$

The system matrix $R \in A^{4\times 6}$ of (1) is defined by:

```
> A:=DefineOreAlgebra(diff=[d,t],dual_shift=[sigma[1],x[1]],
> dual_shift=[sigma[2],x[2]],polynom=[t,x[1],x[2]],
> comm=[eta[1],eta[2]]):
```

```
> R:=matrix(4,6,[1,1,-1,-1,0,0,d+eta[1],d-eta[1],-eta[2],
> eta[2],0,0, sigma[1]^2,1,0,0,-sigma[1],0,0,0,1,sigma[2]^2,
> 0,-sigma[2]]);
```

$$R := \begin{bmatrix} 1 & 1 & -1 & -1 & 0 & 0 \\ d + \eta_1 & d - \eta_1 & -\eta_2 & \eta_2 & 0 & 0 \\ {\sigma_1}^2 & 1 & 0 & 0 & -\sigma_1 & 0 \\ 0 & 0 & 1 & {\sigma_2}^2 & 0 & -\sigma_2 \end{bmatrix}$$

Factorization problem We show how to use OreMorphisms for computing a factorization of R of the form $R = L\,S$. We first need to compute the endomorphism ring $\mathrm{end}_A(M)$ of the A-module $M = A^{1\times 6}/(A^{1\times 4}\,R)$ finitely presented by the matrix R.

```
> Endo:=MorphismsConstCoeff(R,R,A):
```

Then, we choose a particular morphism f by selecting the first element P_1 of *Endo[1]* and compute a matrix Q_1 satisfying $R\,P_1 = Q_1\,R$. The latter operation can be performed by means of the *Factorize* procedure of OreModules.

```
> P[1]:=Endo[1,1]; Q[1]:=Factorize(Mult(R,P[1],A),R,A);
```

$$P_1 := \begin{bmatrix} 0 & 0 & \eta_2\,\sigma_2 & \eta_2\,\sigma_2 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \sigma_2\,\eta_1 & \eta_2\,\sigma_2 & 0 & 0 & 0 \\ 0 & -\sigma_2\,\eta_1 & 0 & \eta_2\,\sigma_2 & 0 & 0 \\ 0 & 0 & \eta_2\,\sigma_2\,\sigma_1 & \eta_2\,\sigma_2\,\sigma_1 & 0 & 0 \\ 0 & \eta_1 - {\sigma_2}^2\eta_1 & 0 & 0 & 0 & \eta_2\,\sigma_2 \end{bmatrix}$$

$$Q_1 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ -\eta_2\,\sigma_2\,\eta_1 - \eta_2\,\sigma_2\,d & \eta_2\,\sigma_2 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \eta_2\,\sigma_2 \end{bmatrix}$$

By Theorem 3.1 of T. Cluzeau, A. Quadrat, "Factoring and decomposing a class of linear functional systems", *Linear Algebra and Its Applications*, 428 (2008), 324-381, the matrix S that we are searching for is the one defining the coimage of the endomorphism f of M defined by the previous matrices P_1 and Q_1. So, we compute it using the *CoimMorphism* procedure.

```
> S:=CoimMorphism(R,R,P[1],Q[1],A)[1];
```

$$S := \begin{bmatrix} 1 & 0 & -1 & -1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & \sigma_1 & \sigma_1 & -1 & 0 \\ 0 & 0 & 1 & {\sigma_2}^2 & 0 & -\sigma_2 \\ 0 & 0 & -d + \eta_2 - \eta_1 & -d - \eta_2 - \eta_1 & 0 & 0 \end{bmatrix}$$

The matrix L such that $R = L\,S$ can be obtained by right factoring R by S.

```
> L:=Factorize(R,S,A);
```

$$L := \begin{bmatrix} 1 & 1 & 0 & 0 & 0 & 0 \\ d + \eta_1 & d - \eta_1 & 0 & 1 & 0 & 0 \\ {\sigma_1}^2 & 1 & \sigma_1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix}$$

We note that choosing another endomorphism of M, i.e., another element of *Endo*[1], would lead to another factorization of the matrix R.

Reduction problem We use the package OreMorphisms to reduce the matrix R, i.e., to find an equivalent matrix with a block-triangular form. Theorem 3.2 of T. Cluzeau, A. Quadrat, "Factoring and decomposing a class of linear functional systems", *Linear Algebra and Its Applications*, 428 (2008), 324-381, this can be done using an endomorphism of M defined by a pair of matrices P and Q provided that the A-modules $\ker_A(.P)$, $\mathrm{coim}_A(.P)$, $\ker_A(.Q)$ and $\mathrm{coim}_A(.Q)$ are free. We use the library OreModules to check that these properties are fulfilled and use a heuristic method to compute bases of those free A-modules. We then form the matrices U and V as defined in Theorem 3.2 of T. Cluzeau, A. Quadrat, "Factoring and decomposing a class of linear functional systems", *Linear Algebra and Its Applications*, 428 (2008), 324-381. We note that we generally need to use the package QuillenSuslin to compute bases of free modules over a commutative polynomial ring (see A. Fabiańska, A. Quadrat, "Applications of the Quillen-Suslin theorem in multidimensional systems theory", chapter of the book *Gröbner Bases in Control Theory and Signal Processing*, H. Park and G. Regensburger (Eds.), Radon Series on Computation and Applied Mathematics 3, de Gruyter publisher, 2007, 23-106).

```
> U1:=SyzygyModule(P[1],A): EU:=Exti(Involution(U1,A),A,1):
> U2:=LeftInverse(EU[3],A): U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q[1],A): EV:=Exti(Involution(V1,A),A,1):
> V2:=LeftInverse(EV[3],A): V:=stackmatrix(V1,V2);
```

$$U := \begin{bmatrix} 1 & 0 & -1 & -1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & \sigma_1 & \sigma_1 & -1 & 0 \\ 0 & 0 & -1 & -1 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad V := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Then, we can compute the reduction $V\,R\,U^{-1}$ of the matrix R:

```
> R_red:=Mult(V,R,LeftInverse(U,A),A);
```

$$R\_red := \begin{bmatrix} 1 & 1 & 0 & 0 & 0 & 0 \\ {\sigma_1}^2 & 1 & \sigma_1 & 0 & 0 & 0 \\ d+\eta_1 & d-\eta_1 & 0 & -\eta_1-\eta_2-d & -2\,\eta_2 & 0 \\ 0 & 0 & 0 & -{\sigma_2}^2 & 1-{\sigma_2}^2 & -\sigma_2 \end{bmatrix}$$

This reduction can be obtained using the *HeuristicReduction* procedure.

```
> HeuristicReduction(R,P[1],A)[1];
```

$$\begin{bmatrix} 1 & 1 & 0 & 0 & 0 & 0 \\ {\sigma_1}^2 & 1 & \sigma_1 & 0 & 0 & 0 \\ d+\eta_1 & d-\eta_1 & 0 & -\eta_1-\eta_2-d & -2\,\eta_2 & 0 \\ 0 & 0 & 0 & -{\sigma_2}^2 & 1-{\sigma_2}^2 & -\sigma_2 \end{bmatrix}$$

Decomposition problem We now show how to use the package OreMorphisms to decompose the differential time-delay linear system (1), i.e., to find an equivalent system defined by a block-diagonal matrix. To achieve this decomposition, we first need to compute idempotent endomorphisms of M that are defined by idempotent matrices P and Q i.e., $R\,P = Q\,R$, $P^2 = P$ and $Q^2 = Q$. A way to do that is to use the procedure *IdempotentsMatConstCoeff* of OreMorphisms. We need to specify the total order in d, σ_1 and σ_2 of the idempotent matrix P, a piece of information which is specified by the fourth entry of the procedure. We first start by searching for idempotents of M defined by constant matrices.

```
> Idem_order0:=IdempotentsMatConstCoeff(R,Endo[1],A,0)[1];
```

$$Idem\_order0 := \left[\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & -1 & 1 & 1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \right]$$

We choose the non-trivial idempotent, i.e., the second entry of $Idem\_order0$:

```
> P[2]:=Idem_order0[2]; Q[2]:=Factorize(Mult(R,P[2],A),R,A);
```

$$P_2 := \begin{bmatrix} 0 & -1 & 1 & 1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad Q_2 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ -d-\eta_1 & 1 & 0 & 0 \\ -{\sigma_1}^2 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

The fact that $P_2^2 = P_2$ and $Q_2^2 = Q_2$ imply that the A-modules $\ker_A(.P_2)$, $\ker_A(.Q_2)$, $\operatorname{im}_A(.P_2) = \ker_A(.(I_6 - P_2))$ and $\operatorname{im}_A(.Q_2) = \ker_A(.(I_4 - Q_2))$ are projective, and thus, free by the Quillen-Suslin theorem. We need to compute bases of those free A-modules. We then form the matrices U and V as explained in Theorem 4.2 of T. Cluzeau, A. Quadrat, "Factoring and decomposing a class of linear functional systems", *Linear Algebra and Its Applications*, 428 (2008), 324-381.

```
> U1:=SyzygyModule(P[2],A): U2:=SyzygyModule(evalm(1-P[2]),A):
> U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q[2],A): V2:=SyzygyModule(evalm(1-Q[2]),A):
> V:=stackmatrix(V1,submatrix(V2,[1, 2, 4],1..4));
```

$$U := \begin{bmatrix} 1 & 1 & -1 & -1 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad V := \begin{bmatrix} 1 & 0 & 0 & 0 \\ d+\eta_1 & -1 & 0 & 0 \\ {\sigma_1}^2 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Now, we can compute the corresponding decomposition $V\,R\,U^{-1}$ of R:

```
> R_dec:=Mult(V,R,LeftInverse(U,A),A);
```

$$R\_dec := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 2\,\eta_1 & -d+\eta_2-\eta_1 & -d-\eta_2-\eta_1 & 0 & 0 \\ 0 & {\sigma_1}^2-1 & -{\sigma_1}^2 & -{\sigma_1}^2 & \sigma_1 & 0 \\ 0 & 0 & 1 & {\sigma_2}^2 & 0 & -\sigma_2 \end{bmatrix}$$

We can now try to decompose the second diagonal block matrix S of R_dec:

```
> S:=submatrix(R_dec,2..4,2..6):
```

We apply the same technique as above: compute the endomorphism ring of the A-module $N = A^{1\times 5}/(A^{1\times 3}\,S)$ finitely presented by S, find one idempotent defined by idempotent matrices, compute bases of the free A-modules defined by their kernels and images, form the corresponding unimodular matrices and deduce the decomposition of S.

```
> Endo1:=MorphismsConstCoeff(S,S,A):
> Idem1_order0:=IdempotentsMatConstCoeff(S,Endo1[1],A,0)[1];
```

$$Idem1\_order0 := \left[\begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix} \right]$$

We do not obtain a non-trivial idempotent of order 0 by means of the *IdempotentsMatConstCoeff* procedure. Hence, we can try another technique which searches for idempotents which are obtained by homotopies from the trivial idempotent id_N defined by $P_3 = I_5$ and $Q_3 = I_3$, i.e., $S\,P_3 = Q_3\,S$.

```
> P[3]:=diag(1$5):  Q[3]:=diag(1$3):  Z[3]:=matrix(5,3,[0$15]):
```

We then need to solve the algebraic Riccati equation $\Lambda\,S\,\Lambda + \Lambda = 0$:

```
> Mu:=RiccatiConstCoeff(S,P[3],Q[3],Z[3],A,0,alpha):
```

We choose one solution Λ_1 of the previous algebraic Riccati equation:

```
> Lambda[1]:=subs({b321=0,b521=0},Mu[1,2]);
```

$$\Lambda_1 := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

We get a non-trivial idempotent defined by the following matrices P_4 and Q_4:

```
> P[4]:=simplify(evalm(P[3]+Mult(Lambda[1],S,A)));
> Q[4]:=simplify(evalm(Q[3]+Mult(S,Lambda[1],A)));
```

$$P_4 := \begin{bmatrix} \sigma_1{}^2 & -\sigma_1{}^2 & -\sigma_1{}^2 & \sigma_1 & 0 \\ \sigma_1{}^2-1 & -\sigma_1{}^2+1 & -\sigma_1{}^2 & \sigma_1 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad Q_4 := \begin{bmatrix} 1 & \eta_1 - d + \eta_2 & 0 \\ 0 & 0 & 0 \\ 0 & 1 & 1 \end{bmatrix}$$

We now compute bases of the free A-modules $\ker_A(.P_4)$, $\ker_A(.Q_4)$, $\mathrm{im}_A(.P_4) = \ker_A(.(I_5 - P_4))$ and $\mathrm{im}_A(.Q_4) = \ker_A(.(I_3 - Q_4))$ and we get the following two unimodular matrices X and Y:

```
> X1:=SyzygyModule(P[4],A): X2:=SyzygyModule(evalm(1-P[4]),A):
> X:=stackmatrix(X1,X2);
> Y1:=SyzygyModule(Q[4],A): Y2:=SyzygyModule(evalm(1-Q[4]),A):
> Y:=stackmatrix(Y1,Y2);
```

$$X := \begin{bmatrix} \sigma_1{}^2-1 & -\sigma_1{}^2 & -\sigma_1{}^2 & \sigma_1 & 0 \\ 1 & -1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad Y := \begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & d - \eta_2 - \eta_1 \\ 0 & 1 & 1 \end{bmatrix}$$

Then, we obtain the following decomposition $Y\,S\,X^{-1}$ of the matrix S:

```
> S_dec:=Mult(Y,S,LeftInverse(X,A),A);
```

$$S\_dec := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 2\,\eta_1 & -d - \eta_2 - \eta_1 + \sigma_2{}^2\,d - \eta_2\,\sigma_2{}^2 - \sigma_2{}^2\,\eta_1 & 0 & (\eta_1 - d + \eta_2)\,\sigma_2 \\ 0 & \sigma_1{}^2 - 1 & -\sigma_1{}^2 + \sigma_2{}^2 & \sigma_1 & -\sigma_2 \end{bmatrix}$$

We continue by considering the second diagonal block matrix T of S_dec:

```
> T:=submatrix(S_dec,2..3,2..5):
```

We apply the same technique as above:

```
> P[5]:=diag(1$4):  Q[5]:=diag(1$2):  Z[5]:=matrix(4,2,[0$8]):
```

We compute the solutions of the Riccati equation $\Lambda\, T\, \Lambda + \Lambda = 0$:

```
> Mu1:=RiccatiConstCoeff(T,P[5],Q[5],Z[5],A,0,alpha):
```

We choose one solution Λ_2 of the previous algebraic Riccati equation:

```
> Lambda[2]:=subs({b311=0},Mu1[1,1]);
```

$$\Lambda_2 := \begin{bmatrix} -1/(2\,\eta_1) & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \end{bmatrix}$$

Hence, we get an idempotent of the endomorphism ring of the A-module finitely presented by T defined by the following matrices P_6 and Q_6:

```
> P[6]:=simplify(evalm(P[5]+Mult(Lambda[2],T,A)));
> Q[6]:=simplify(evalm(Q[5]+Mult(T,Lambda[2],A)));
```

$$P_6 = \begin{bmatrix} 0 & 1/2\,\frac{\eta_1+\eta_2+d-{\sigma_2}^2\,d+\eta_2\,{\sigma_2}^2+{\sigma_2}^2\eta_1}{\eta_1} & 0 & -1/2\,\frac{(\eta_1-d+\eta_2)\sigma_2}{\eta_1} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$Q_6 = \begin{bmatrix} 0 & 0 \\ -1/2\,\frac{{\sigma_1}^2-1}{\eta_1} & 1 \end{bmatrix}$$

We now compute bases of the free A-modules $\ker_A(.P_6)$, $\ker_A(.Q_6)$, $\mathrm{im}_A(.P_6) = \ker_A(.(I_4 - P_6))$ and $\mathrm{im}_A(.Q_6) = \ker_A(.(I_2 - Q_6))$ and we obtain the following unimodular matrices G and H:

```
> G1:=SyzygyModule(P[6],A): G2:=SyzygyModule(evalm(1-P[6]),A):
> G:=stackmatrix(G1,G2);
> H1:=SyzygyModule(Q[6],A): H2:=SyzygyModule(evalm(1-Q[6]),A):
> H:=stackmatrix(H1,H2);
```

$$G := \begin{bmatrix} 2\,\eta_1 & -d-\eta_2-\eta_1+{\sigma_2}^2\,d-\eta_2\,{\sigma_2}^2-{\sigma_2}^2\eta_1 & 0 & \sigma_2\,\eta_1-\sigma_2\,d+\eta_2\,\sigma_2 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$H := \begin{bmatrix} 1 & 0 \\ {\sigma_1}^2-1 & -2\,\eta_1 \end{bmatrix}$$

Then, we obtain the following decomposition $H\, T\, G^{-1}$ of the matrix T:

```
> T_dec:=Mult(H,T,LeftInverse(G,A),A);
```

$$T\_dec :=$$

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & \left((-\eta_1 + d - \eta_2)\,\sigma_2{}^2 + \eta_1 - d - \eta_2\right)\sigma_1{}^2 + (-d - \eta_1 + \eta_2)\,\sigma_2{}^2 + d + \eta_2 + \eta_1 & -2\,\eta_1\,\sigma_1 & (\eta_1 - d + \eta_2)\,\sigma_2\,\sigma_1{}^2 + (d - \eta_2 + \eta_1)\,\sigma_2 \end{bmatrix}$$

From the previous three invertible transformations, we can deduce the unimodular matrices that perform all this decomposition process in one step:

```
> W[1]:=Mult(diag(1,1,G),diag(1,X),U,A):
> W[2]:=Mult(diag(1,1,H),diag(1,Y),V,A):
```

The system matrix R is equivalent to the matrix $L = W_2\,R\,W_1^{-1}$.

```
> L:=Mult(W[2],R,LeftInverse(W[1],A),A)):
```

The matrix L has then the form

```
> ShapeOfMatrix(L);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & * & * & * \end{bmatrix}$$

where the stars * denote non-trivial elements of A respectively defined by:

```
> collect(L[4,4],{d,sigma[1],sigma[2]});
```

$$\left((-\eta_1 + d - \eta_2)\,\sigma_2{}^2 + \eta_1 - d - \eta_2\right)\sigma_1{}^2 + (-d - \eta_1 + \eta_2)\,\sigma_2{}^2 + d + \eta_2 + \eta_1$$

```
> collect(L[4,5],{d,sigma[1],sigma[2]});
```

$$-2\,\eta_1\,\sigma_1$$

```
> collect(L[4,6],{d,sigma[1],sigma[2]});
```

$$(\eta_1 - d + \eta_2)\,\sigma_2\,\sigma_1{}^2 + (d - \eta_2 + \eta_1)\,\sigma_2$$

The entries of the last row of L can be reduced by means of elementary column operations. Hence, if we consider the following unimodular matrix

$$J :=$$

$$\begin{bmatrix} 1 & 1 & -1 & -1 & 0 & 0 \\ 0 & \sigma_1{}^2 - 1 & -\sigma_1{}^2 & -\sigma_1{}^2 & \sigma_1 & 0 \\ 0 & 2\,\eta_1 & -2\,\eta_1 & -\eta_1 - \eta_2 - d + \sigma_2{}^2\,d - \sigma_2{}^2\,\eta_1 - \eta_2\,\sigma_2{}^2 & 0 & -(d - \eta_1 - \eta_2)\,\sigma_2 \\ 0 & 0 & 0 & 1 - \sigma_2{}^2 & 0 & \sigma_2 \\ 0 & 0 & 0 & \sigma_1\left(\sigma_2{}^2\,d - \sigma_2{}^2\,\eta_1 - \eta_2\,\sigma_2{}^2 - d - \eta_2 + \eta_1\right) & -2\,\eta_1 & -\sigma_2\,\sigma_1\,(d - \eta_1 - \eta_2) \\ 0 & 0 & 0 & 2\,\sigma_2\,\eta_2 & 0 & -2\,\eta_2 \end{bmatrix}$$

obtained from W_1 by means of elementary operations (see the corresponding Maple worksheet), we finally get the following simpler decomposition $W_2\,R\,J^{-1}$ of R:

```
> R_final:=Mult(W[2],R,LeftInverse(J,A),A);
```

$$R\_final = \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & d + \eta_1 + \eta_2 & \sigma_1 & \sigma_2 \end{bmatrix}$$

Hence, the differential time-delay system (1) formed by 4 equations in 6 unknowns is equivalent to the following sole equation in 3 unknowns:

$$\dot{x}_1(t) + (\eta_1 + \eta_2)\, x_1(t) + x_2(t - h_1) + x_3(t - h_2) = 0. \tag{2}$$

Using the simple form of (2), we can easily study its structural properties (e.g., controllability, parametrizability, flatness, π-freeness, stability, stabilizability), and thus, those of (1). In particular, we obtain that (2), and thus, (1) is controllable, parametrizable, σ_1-free and σ_2-free (see F. Chyzak, A. Quadrat, D. Robertz, "Effective algorithms for parametrizing linear control systems over Ore algebras", *Appl. Algebra Engrg. Comm. Comput.*, 16 (2005), 319-376, and M. Fliess, H. Mounier, "Controllability and observability of linear delay systems: an algebraic approach", *ESAIM: Control, Optimisation and Calculus of Variations*, 3 (1998), 301-314, for the corresponding definitions). Parametrizations of (1) can directly be obtained from the ones of (2) by means of the matrix J^{-1} (see the previous references). System (2) admits an unstable pole at $-(\eta_1 + \eta_2)$, where the η_i's are two positive parameters of (1). Its stabilizability can be studied using, e.g., A. Quadrat, "The fractional representation approach to synthesis problems: an algebraic analysis viewpoint. Part II: Internal stabilization", *SIAM J. Control & Optimization*, 42 (2003), 300-320.